
1. Recall the maximum segment sum problem. What if we want to compute the maximum sum of non-empty
segments?

(a) How would you write the specification? Does the specification still make sense with N being constrained
only by 0 ⩽ N?

Solution: The specification could be:

con N : Int {0 ⩽ N}
con f : array [0..N) of Int
S
{r = ⟨↑ p q : 0 ⩽ p< q ⩽ N : sum p q⟩} ,

where the definition of sum is unchanged:

sum p q = ⟨Σi : p ⩽ i < q : f [i]⟩ .

When N = 0, 0 ⩽ p< q ⩽ N reduces to False and r should be −∞. The specification is fine if −∞ is a
value allowed in our program.

(b) Derive a program solving the problem, assuming −∞ to be the identity of (↑).

Solution: Like in the handouts, we start with

P0 ≡ r = ⟨↑ p q : 0 ⩽ p< q ⩽ n : sum p q⟩ ,
Q ≡ 0 ⩽ n ⩽ N ,

and use N − n as bound.
To find out what we can do with r before n := n + 1, we calculate:

⟨↑ p q : 0 ⩽ p< q ⩽ n : sum p q⟩[n\n + 1]
= ⟨↑ p q : 0 ⩽ p< q ⩽ n + 1 : sum p q⟩
= { split off q = n + 1 (safe since 0 ⩽ n) }
⟨↑ p q : 0 ⩽ p< q ⩽ n : sum p q⟩ ↑ ⟨↑ p : 0 ⩽ p< n + 1 : sum p (n + 1)⟩ .

Therefore we add another invariant:

P1 ≡ s = ⟨↑ p : 0 ⩽ p< n : sum p n⟩ .

Variables r , s, n can be initialised by −∞,−∞, 0. To find out how to update s, we calculate:

⟨↑ p : 0 ⩽ p< n : sum p n⟩[n\n + 1]
= ⟨↑ p : 0 ⩽ p< n + 1 : sum p (n + 1)⟩
= { split off p = n (safe since 0 ⩽ n) }
⟨↑ p : 0 ⩽ p< n : sum p (n + 1)⟩ ↑ sum n (n + 1)

= { definition of sum, one-point rule }
⟨↑ p : 0 ⩽ p< n : sum p (n + 1)⟩ ↑ f [n]

= { split off i = n in definition of sum, (safe since 0 ⩽ n) }
⟨↑ p : 0 ⩽ p< n : sum p n + f [n]⟩ ↑ f [n]

= { p not free in f [n] }
(⟨↑ p : 0 ⩽ p< n : sum p n⟩ + f [n]) ↑ f [n] .

The constructed program is:

con N : Int {0 ⩽ N}
con f : array [0..N) of Int
var r , s, n : Int
r , s, n :=−∞,−∞, 0
{P0 ∧ P1 ∧ Q, bnd : N − n}
do n ̸= N →
s := (s + f [n]) ↑ f [n]
r := r ↑ s
n := n + 1

od
{r = ⟨↑ p q : 0 ⩽ p< q ⩽ N : sum p q⟩}

2. Let n :Nat . Trying explaining why the following “splitting off” step is wrong by trying out the range calculation.

⟨Σi : n + 1 ⩽ i < n + 1 : f [i]⟩
= { splitting off i = n }
⟨Σi : n + 1 ⩽ i < n : f [i]⟩ + f [n] .

Solution: Let’s calculate:

n + 1 ⩽ i < n + 1
= n + 1 ⩽ i ∧ (i < n ∨ i = n)
= n + 1 ⩽ i < n ∨ (n + 1 ⩽ i ∧ i = n)
= n + 1 ⩽ i < n ∨ False
= n + 1 ⩽ i < n .

The term n + 1 ⩽ i ∧ i = n simplifies to False, not i = n, which the “splitting off” step might expect.

What we do have is

⟨Σi : n + 1 ⩽ i < n + 1 : f [i]⟩
= ⟨Σi : n + 1 ⩽ i < n : f [i]⟩ .

In fact, both expressions evalulate to 0.

3. Derive a solution for:

con N : Int{N ⩾ 0}; a : array [0..N)of Int
var r : Int
S
{r = ⟨ ↑ i, j : 0 ⩽ i < j < N : a[i]− a[j] ⟩} .

Solution: Replace constant N by variable n, and use a loop that increments n in each step. Use the follow-
ing P as a candidate of the loop invariant

P ≡ r = ⟨ ↑ i, j : 0 ⩽ i < j < n : a[i]− a[j] ⟩.

To find out how to update r such that we may increment n, we calculate (assuming 0 ⩽ n):

⟨ ↑ i, j : 0 ⩽ i < j < n + 1 : a[i]− a[j] ⟩
= { since 0 ⩽ n, split off j = n }

⟨ ↑ i, j : 0 ⩽ i < j < n : a[i]− a[j] ⟩ ↑ ⟨ ↑ i : 0 ⩽ i < n : a[i]− a[n] ⟩
= { n not bounded }

⟨ ↑ i, j : 0 ⩽ i < j < n : a[i]− a[j] ⟩ ↑ (⟨ ↑ i : 0 ⩽ i < n : a[i] ⟩ − a[n]).

So we strengthen the invariant by adding a variable s satisfying

Q ≡ s = ⟨ ↑ i : 0 ⩽ i < n : a[i] ⟩.

The invariant is P ∧ Q ∧ 0 ⩽ n ⩽ N .

The program:

con N : Int {0 ⩽ N}
con a : array [0..N) of Int
var r , s, n : Int
r , s, n :=−∞,−∞, 0 -- Pf0
{P ∧ Q ∧ 0 ⩽ n ⩽ N , bnd : N − n} -- Pf1
do n ̸= N →

r , s, n := r ↑ (s − a[n]), s ↑ a[n], n + 1 -- Pf2
od
{r = ⟨↑ i j : 0 ⩽ i < j < N : a[i]− a[j]⟩} -- Pf3

Here I am omitting other proofs and presenting only Pf2:

Pf2

(P ∧ Q ∧ 0 ⩽ n ⩽ N)[r , s, n\r ↑ (s − a[n]), s ↑ a[n], n + 1]

≡ r ↑ (s − a[n]) = ⟨ ↑ i, j : 0 ⩽ i < j < n + 1 : a[i]− a[j] ⟩ ∧
s ↑ a[n] = ⟨ ↑ i : 0 ⩽ i < n + 1 : a[i] ⟩ ∧ 0 ⩽ n + 1 ⩽ N

⇐ { split off i = n }
r ↑ (s − a[n]) = ⟨ ↑ i, j : 0 ⩽ i < j < n : a[i]− a[j] ⟩ ↑ (⟨ ↑ i : 0 ⩽ i < n : a[i] ⟩ − a[n]) ∧
s ↑ a[n] = ⟨ ↑ i : 0 ⩽ i < n : a[i] ⟩ + a[n] ∧ 0 ⩽ n < N

⇐ r = ⟨ ↑ i, j : 0 ⩽ i < j < n : a[i]− a[j] ⟩ ∧
s = ⟨ ↑ i : 0 ⩽ i < n : a[i] ⟩ ∧ 0 ⩽ n ⩽ N ∧ n ̸= N

≡ P ∧ Q ∧ 0 ⩽ n ⩽ N ∧ n ̸= N .

4. Derive a solution for:

con N : Int{N ⩾ 1}; a : array [0..N)of Int
var r : Int
S
{r = ⟨ #i, j : 0 ⩽ i < j < N : a[i]× a[j] ⩾ 0 ⟩} .

Solution: Replace constant N by variable n, and use a loop that increments n in each step. Let

P ≡ r = ⟨ #i, j : 0 ⩽ i < j < n : a[i]× a[j] ⩾ 0 ⟩.

We first attempt to use P ∧ 0 ⩽ n ⩽ N as the invariant and, apparently, N − n as the bound. To find out
how to update r such that we may increment n, we calculate (assuming 0 ⩽ n):

⟨ #i, j : 0 ⩽ i < j < n + 1 : a[i]× a[j] ⩾ 0 ⟩
= { since 0 ⩽ n, splitting off j = n }

⟨ #i, j : 0 ⩽ i < j < n : a[i]× a[j] ⩾ 0 ⟩ + ⟨ #i : 0 ⩽ i < n : a[i]× a[n] ⩾ 0 ⟩.

To further simplify ⟨ #i : 0 ⩽ i < n : a[i]× a[n] ⩾ 0 ⟩, we do a case analysis on a[n]:

⟨ #i : 0 ⩽ i < n : a[i]× a[n] ⩾ 0 ⟩ = ⟨ #i : 0 ⩽ i < n : a[i] ⩾ 0 ⟩, if a[n] > 0;
n, if a[n] = 0;
⟨ #i : 0 ⩽ i < n : a[i] ⩽ 0 ⟩, if a[n] < 0.

Thus we strengthen the invariant by adding two more variables:

Q1 ≡ s1 = ⟨ #i : 0 ⩽ i < n : a[i] ⩾ 0 ⟩ ,

Q2 ≡ s2 = ⟨ #i : 0 ⩽ i < n : a[i] ⩽ 0 ⟩ .

The invariant is P ∧ Q1 ∧ Q2 ∧ 0 ⩽ n ⩽ N .

The program:

con N : Int{N ⩾ 1}; a : array [0..N)of Int
var r , s1, s2, n : Int

r , s1, s2, n := 0, 0, 0, 0
{P ∧ Q1 ∧ Q2 ∧ 0 ⩽ n ⩽ N , bnd : N − n}
do n < N →
if a[n] > 0 → r , s1, n := r + s1, s1 + 1, n + 1
| a[n] = 0 → r , s1, s2, n := r + n, s1 + 1, s2 + 1, n + 1
| a[n] < 0 → r , s2, n := r + s2, s2 + 1, n + 1
fi

od
{r = ⟨ #i j : 0 ⩽ i < j < N : a[i]× a[j] ⩾ 0 ⟩} .

5. Consider again the maximum segment sum problem and its derivation in the handouts. Since the computation
of r requires the value of the subterm ⟨↑ p : 0 ⩽ p ⩽ n + 1 : sum p (n + 1)⟩, some would think it makes more
sense to use the following loop invariant P0 ∧ P1 ∧ Q, where

P0 ≡ r = ⟨↑ p q : 0 ⩽ p ⩽ q ⩽ n : sum p q⟩) ,
P1 ≡ s = ⟨↑ p : 0 ⩽ p ⩽ n + 1 : sum p (n + 1)⟩ ,
Q ≡ 0 ⩽ n ⩽ N .

What situation could you run into, if you try to construct a program using the invariant above? What if the
array is non-empty, that is, 1 ⩽ N?

Solution: With 0 ⩽ N we would run into problem initialising s. The initialisation may look like:

r , s, n := 0, ?, 0
{P0 ∧ P1 ∧ Q}

and the value of s should be

P1[r , s, n\0, ?, 0]
≡ ? = ⟨↑ p : 0 ⩽ p ⩽ 1 : sum p 1⟩
≡ ? = sum 0 1 + sum 1 1
≡ ? = f [0] + 0
≡ ? = f [0] .

However, f [0] does not have a value when N = 0.

When we have 1 ⩽ N instead of 0 ⩽ N we will be able to initialize the variables by r , s, n := 0, f [0], 0. When
we construct the loop body, knowing that P0[n\n + 1] ≡ r = r ↑ ⟨↑ p : 0 ⩽ p ⩽ (n + 1) : sum p (n + 1)⟩, the
constructed loop body would be:

{P0 ∧ P1 ∧ 0 ⩽ n ⩽ N ∧ n ̸= N}
r := r ↑ s
{P0[n\n + 1] ∧ P1 ∧ 0 ⩽ n + 1 ⩽ N}
s := (s + f [n + 1]) ↑ 0
{(P0 ∧ P1 ∧ 0 ⩽ n ⩽ N)[n\n + 1]}
n := n + 1
{P0 ∧ P1 ∧ 0 ⩽ n ⩽ N}

Note that the order of assignments is different from that in the handouts.

However, upon termination (that is, n = N) we would need to establish

s = ⟨↑ p : 0 ⩽ p ⩽ N + 1 : sum p (N + 1)⟩ ,

which we cannot do because f [N] is not defined.

It is possible to fix all these: for example, terminate the loop one step earlier and do some post processing,
and put the entire loop under an if to ensure that 1 ⩽ N . The resulting program would not be as clean as
the one in the handouts, though.

